
Umrechnung der Parameter der GF-Verteilung für Unterverteilungen

Verteilung GF-Verteilung
Log-logistische 
Verteilung

Log-Normal- 
Verteilung

Weibull-  
Verteilung

Gamma-
Verteilung

 @RISK-NAME
RISK 
LOGLOGISTIC RISK LOGNORM2

RISK       
WEIBULL RISK       GAMMA

Parameter der GF-Verteilung (rote Angaben sind fest)
Parameter

m1= 9.11183 1 ∞ 1 11.74583

m2= 9.70326 1 ∞ ∞ ∞

λ= 0.00049 0.00049 0.00049 0.00042 0.00047

p= 1.61247 5.95789 3.40278 3.36433 1

Parameter der @RISK-Verteilungen
Parameter

Shape α= 5.9579 3.3643 11.7458

Scale β= 2042.9009 2363.5074 181.4506

Mean μ= 7.6213

Std-Dev. σ= 0.2939

(Umrechnung für Lognormalverteilung) ω2= 1.0902

 @RISK-Befehl 2140.7343 2131.3079 2122.1800 2131.2873

Kennzahlen

Arithm. Mittel E(X) 2131.0145 2140.7343 2131.3079 2122.1800 2131.2873

Funktionen

Beobachtung x= 2000.0000 2000.0000 2000.0000 2000.0000 2000.0000

Dichtefunktion f(x)= 0.0007 0.0007 0.0005 0.0007

Verteilungsfunktion F(X)= 0.4711 0.4684 0.4723 0.4346 0.4532

Wahrscheinlichkeit p 0.5 0.5 0.5 0.5 0.5

Quantil q(p)= 2042.8843 2042.9009 2041.2329 2119.5542 2071.1185

Xprob-Anhang

Berechnungen: Zelle  =Formel

GF-Verteilung:
Eingaben:
1. Parameter: m1 C9
2. Parameter: m2 C10
3. Parameter: λ C11
4. Parameter: p C12
Beobachtung: x C30
Wahrscheinlichkeit: p C34

Ausgaben:
Arithm. Mittelwert: E(X) C26= EXP(LN($C$10/$C$9)/$C$12)*EXP(GAMMALN($C$9+1/$C$12))

 *EXP(GAMMALN($C$10-1/$C$12))/$C$11/EXP(GAMMALN($C$9))/EXP(GAMMALN($C$10))
Verteilungsfunktion: F(x) C32= 1-FVERT(EXP(LN($C$11*$C$30)*$C$12);2*$C$9;2*$C$10)
Quantil: q(p)= C35= EXP(LN(FINV($C$34;2*$C$9;2*$C$10))/$C$12)/$C$11
Beachte: Für große Werte von m1 bzw. m2 sind nicht alle Berechnungen mit EXCEL möglich.

Log-logistische Verteilung:
Eingaben:
1. GF-Parameter: λ D11
2. GF-Parameter: p D12
Beobachtung: x D30
Wahrscheinlichkeit: p D34

Ausgaben:
1. @RISK-Parameter: α D16= $D$12
2. @RISK-Parameter: β D17= 1/$D$11
 @RISK-Befehl: D22= RiskLoglogistic(0;$D$17;$D$16)
Arithm. Mittelwert: E(X) D26= EXP(GAMMALN(1+1/$D$12))*EXP(GAMMALN(1-1/$D$12))/$D$11
Dichtefunktion: f(x) D31= $D$11*$D$12*EXP(LN($D$11*$D$30)*($D$12-1))/(1+EXP(LN($D$11*$D$30)*($D$12)))^2
Verteilungsfunktion: F(x) D32= EXP(LN($D$11*$D$30)*$D$12)/(1+EXP(LN($D$11*$D$30)*$D$12))
Quantil: q(p)= D35= EXP(LN($D$34/(1-$D$34))/$D$12)/$D$11

Log-Normalverteilung:
Eingaben:
1. GF-Parameter: λ E11
2. GF-Parameter: p E12
Beobachtung: x E30
Wahrscheinlichkeit: p E34

Ausgaben:
1. @RISK-Parameter: μ E18=  -LN($E$11)
2. @RISK-Parameter: σ E19= 1/$E$12
Umrechnung: ω2 E20= EXP($E$19^2)
 @RISK-Befehl: E22= RiskLognorm2($E$18;$E$19)
Arithm. Mittelwert: E(X) E26= EXP($E$18)*WURZEL($E$20)
Dichtefunktion: f(x) E31= NORMVERT( (LN($E$30)-$E$18)/$E$19;0;1;0)/$E$30/$E$19
Verteilungsfunktion: F(x) E32= NORMVERT( (LN($E$30)-$E$18)/$E$19;0;1;1)
Quantil: q(p)= E35= EXP($E$19*NORMINV($E$34;0;1)+$E$18)

Weibull-Verteilung:
Eingaben:
1. GF-Parameter: λ F11
2. GF-Parameter: p F12
Beobachtung: x F30
Wahrscheinlichkeit: p F34

Ausgaben:
1. @RISK-Parameter: α F16= $F$12
2. @RISK-Parameter: β F17= 1/$F$11
 @RISK-Befehl: F22= RiskWeibull($F$16;$F$17)
Arithm. Mittelwert: E(X) F26= $F$17*EXP(GAMMALN(($F$16+1)/$F$16))
Dichtefunktion: f(x) F31= WEIBULL($F$30;$F$16;$F$17;0)
Verteilungsfunktion: F(x) F32= WEIBULL($F$30;$F$16;$F$17;1)
Quantil: q(p)= F35= $F$17*EXP( LN(LN(1/(1-$F$34)))/$F$16)

Gamma-Verteilung:
Eingaben:
1. GF-Parameter: m1 G9
2. GF-Parameter: λ G11
Beobachtung: x G30
Wahrscheinlichkeit: p G34

Ausgaben:
1. @RISK-Parameter: α G16= $G$9
2. @RISK-Parameter: β G17= 1/($G$9*$G$11)
 @RISK-Befehl: G22= RiskGamma($G$16;$G$17)
Arithm. Mittelwert: E(X) G26= $G$16*$G$17
Dichtefunktion: f(x) G31= GAMMAVERT($G$30;$G$16;$G$17;0)
Verteilungsfunktion: F(x) G32= GAMMAVERT($G$30;$G$16;$G$17;1)
Quantil: q(p)= G35= GAMMAINV($G$34;$G$16;$G$17)
Beachte: Für große Werte von m1 sind nicht alle Berechnungen mit EXCEL möglich.

